OEIS A362670

RICHARD J. MATHAR

Abstract. Bounds of the search for solutions of the side lengths a and c for isosceles triangles corresponding to integer radii r in [2, A362670] are developped.

1. Parameters

Sequence A362670 is concerned with isosceles triangles with two sides of length a, one side of length c with the constraint $a < c$. The radius of the incircle is

$$r = \frac{c}{2}\sqrt{\frac{2a-c}{2a+c}}. \tag{1}$$

We search for solutions r with three positive integers r, a and c.

Remark 1. *There are no integer solutions for the equilateral triangle ($a = c$), because then (1) requires $r = c/(2\sqrt{3})$, but $\sqrt{3}$ is an irrational number [1, 3].*

2. Regions

To keep the value of r in (1) positive, the numerator in the square root must be positive, and combined with the constraint by the definition of the sequence

$$c/2 < a < c. \tag{2}$$

Squaring (1) gives

$$4(2a+c)r^2 = c^2(2a-c). \tag{3}$$

This is a linear equation for a:

$$a = \frac{c(4r^2+c^2)}{2(c^2-4r^2)} = -\frac{c}{2} + \frac{c^2}{2(c+2r)} + \frac{c^2}{2(c-2r)}. \tag{4}$$

Remark 2. *This means that all c are even: if c were odd, $4r^2+c^2$ were odd, the product $c(4r^2+c^2)$ were odd, and division through 2 could not yield an integer a.*

Equation (4) means that $a(c)$ has a pole at $c = 2r$ and that $\lim_{c\to\infty} a = c/2$. So for constant r the function $a(c)$ starts high above the line $a = c$ right from $c = (2r)^+$, crosses the line $a = c$ somewhere at $c > 2r$, and approaches the line $a = c/2$ from above at large c. [It never crosses the line $a = c/2$ because that would imply $r = 0$ in (1).] The solutions $a(c)$ at fixed r look roughly as follows:

Date: June 27, 2023.



The curve $a(c)$ can therefore be parametrized with

Definition 1.

$$\epsilon \equiv a - c/2, \quad \epsilon > 0. \tag{5}$$

(4) becomes

$$\epsilon = \frac{4cr^2}{(c+2r)(c-2r)}. \tag{6}$$

Since we are searching for integer a and even c in (5), ϵ must be integer, and therefore

$$\epsilon \geq 1. \tag{7}$$

Combined with (6),

$$\frac{4cr^2}{(c+2r)(c-2r)} \geq 1 \rightsquigarrow c \leq 2r^2 + 2r\sqrt{1+r^2}. \tag{8}$$

This is the upper limit for a numerical search of solutions; the lower limit is obtained from the constraint $\epsilon > 0$ in (6):

$$2r < c \leq 2r^2 + 2r\sqrt{1+r^2}. \tag{9}$$

Appendix A. Maple implementation

```
# @param r the positive integer parameter r
# @return True if r is an inradius for an isosceless triangle with 2 sides a,
#   one side c, a and c integer.
# @author Richard J. Mathar
# @since 2023-06-27
isA362670 := proc(r)
        local c,a,cmax ;
        # upper search limit for c according to manuscript
        cmax := floor(2*r^2+2*r*sqrt(1+r^2)) ;

        # loop over possible side lenghts c
```

```
        for c from 2*r+2 to cmax by 2 do
                # associate side length a
                a := c*(4*r^2+c^2)/2/(c^2-4*r^2) ;

                # check that a is integer and in the range set by the sequence defn.
                if type(a,'integer') and a< c then
                        # print solultions (at most one for each r) ?
                        # printf("r=%d c=%d a=%d\n",r,c,a) ;
                        return true ;
                end if;
        end do:
        return false ;
end proc:

# print solutions in b-file style
n := 1:
for r from 3 do
        if isA362670(r) then
                printf("%d %d\n",n,r) ;
                n := n+1 ;
        end if;
end do:
```

References

1. Leonard Euler, *Of square roots, and of irrational numbers resulting from them*, Springer, 1972.
2. O. E. I. S. Foundation Inc., *The On-Line Encyclopedia Of Integer Sequences*, (2023), https://oeis.org/. MR 3822822
3. Ricardo A. Podestá, *A geometric proof that $\sqrt{3}$, $\sqrt{5}$, and $\sqrt{7}$ are irrational*, arXiv:2003.06627 (2020).

URL: http://www.mpia.de/~mathar

Hoeschstr. 7, 52372 Kreuzau, Germany